

Code :

```
*****
; EZ430-F2013 Demo - Timer_A PWM up/down toggle/set mode, DCO SMCLK
;
; Description; This program will generate one PWM output on P4.1/4.2 using
; Timer_A in up-down mode. The value in CCR0 defines the period/2 and the
; value in CCR1 the off-time/2.
;
; SMCLK = MCLK = TACLK = 1 MHz; 1 micro-sec per cycle
;
; The following is required for a PWM period of 20 m-sec and a
; 70% duty cycle:
;
; Period (cycles) = 20 m-sec / (1 u-sec/cycle x 1000 m-sec/u-sec) = 20,000
;
; CCR0 = 20,000 / 2 = 10,000 (period/2)
; CCR1 = 30% of 10,000 = 3,000 (off-time/2)
;
;
;           MSP430F2013
;           -----
;           |           |
;           |           | XIN|-
;           |           | |
; 70% CCR1/TA1<--|P1.2   | XOUT|-
;           |           | |
;           |           | |
; 50% SMCLK<--|P1.4     |
;           |           |
;
; J.Alvarez
; Texas Instruments, Inc
; Jun 2006
; Built with IAR Embedded Workbench Version: 3.40A
;*****
#include    "msp430x20x3.h"

#define    HLF_PERIOD    10000
#define    HLF_OFF       3000
;-----
;           ORG    0F800h           ; Program Start
;-----
RESET      MOV.W    #027eh,SP       ; Initialize stackpointer
StopWDT    MOV.W    #WDTPW+WDTHOLD,&WDTCTL ; Stop WDT

SetupP1    MOV.B    #0x0f,&P1DIR     ; Set P1.0-P1.7 as outputs
           BIS.B    #BIT2+BIT4,&P1SEL ; P1.2 and P1.4 TA/SMCLK otions

SetupP2    BIS.B    #BIT6+BIT7,&P2DIR ; Set P2.6,P2.7 as outputs

           MOV.B    &CALBC1_1MHZ,&BCSCTL1 ; Set range; DCO = 1 MHz
           MOV.B    &CALDCO_1MHZ,&DCOCTL ; Set DCO step + modulation

SetupC0    MOV.W    #HLF_PERIOD,&TACCR0 ; PWM Period/2
SetupC1    MOV.W    #OUTMOD_6,&TACCTL1 ; CCR1 toggle/set
           MOV.W    #HLF_OFF,&TACCR1 ; CCR1 PWM off time/2

SetupTA    MOV.W    #TASSEL_2+MC_3,&TACTL ; SMCLK, updown mode
;
Mainloop   BIS.W    #CPUOFF,SR      ; CPU off
           NOP                    ; Required only for debugger
;
;-----
;           Interrupt Vectors Used MSP430x20xx
;-----
           ORG    0FFFEh           ; MSP430 RESET Vector
           DW    RESET             ;
           END
```